

NAG Library Routine Document

D01GBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D01GBF returns an approximation to the integral of a function over a hyper-rectangular region, using a Monte Carlo method. An approximate relative error estimate is also returned. This routine is suitable for low accuracy work.

2 Specification

```

SUBROUTINE D01GBF(NDIM, A, B, MINCLS, MAXCLS, FUNCTN, EPS, ACC, LENWRK,
1                  WRKSTR, FINEST, IFAIL)

INTEGER          NDIM, MINCLS, MAXCLS, LENWRK, IFAIL
double precision A(NDIM), B(NDIM), FUNCTN, EPS, ACC, WRKSTR(LenWRk),
1                FINEST
EXTERNAL         FUNCTN

```

3 Description

D01GBF uses an adaptive Monte Carlo method based on the algorithm described in Lautrup (1971). It is implemented for integrals of the form:

$$\int_{a_1}^{b_1} \int_{a_2}^{b_2} \cdots \int_{a_n}^{b_n} f(x_1, x_2, \dots, x_n) dx_n \cdots dx_2 dx_1.$$

Upon entry, unless LENWRK has been set to the minimum value $10 \times \text{NDIM}$, the routine subdivides the integration region into a number of equal volume subregions. Inside each subregion the integral and the variance are estimated by means of pseudorandom sampling. All contributions are added together to produce an estimate for the whole integral and total variance. The variance along each co-ordinate axis is determined and the routine uses this information to increase the density and change the widths of the sub-intervals along each axis, so as to reduce the total variance. The total number of subregions is then increased by a factor of two and the program recycles for another iteration. The program stops when a desired accuracy has been reached or too many integral evaluations are needed for the next cycle.

4 References

Lautrup B (1971) An adaptive multi-dimensional integration procedure *Proc. 2nd Coll. Advanced Methods in Theoretical Physics, Marseille*

5 Parameters

- | | | |
|----|--|--------------|
| 1: | NDIM – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of dimensions of the integral. | |
| | <i>Constraint:</i> $\text{NDIM} \geq 1$. | |
| 2: | A(NDIM) – double precision array | <i>Input</i> |
| | <i>On entry:</i> the lower limits of integration, a_i , for $i = 1, 2, \dots, n$. | |
| 3: | B(NDIM) – double precision array | <i>Input</i> |
| | <i>On entry:</i> the upper limits of integration, b_i , for $i = 1, 2, \dots, n$. | |

4: MINCLS – INTEGER

*Input/Output**On entry:* must be set

either to the minimum number of integrand evaluations to be allowed, in which case $\text{MINCLS} \geq 0$;

or to a negative value. In this case, the routine assumes that a previous call had been made with the same parameters NDIM, A and B and with either the same integrand (in which case D01GBF continues calculation) or a similar integrand (in which case D01GBF begins the calculation with the subdivision used in the last iteration of the previous call). See also WRKSTR.

On exit: contains the number of integrand evaluations actually used by D01GBF.

5: MAXCLS – INTEGER

Input

On entry: the maximum number of integrand evaluations to be allowed. In the continuation case this is the number of new integrand evaluations to be allowed. These counts do not include zero integrand values.

Constraints:

$\text{MAXCLS} > \text{MINCLS}$;
 $\text{MAXCLS} \geq 4 \times (\text{NDIM} + 1)$.

6: FUNCTN – *double precision* FUNCTION, supplied by the user.*External Procedure*

FUNCTN must return the value of the integrand f at a given point.

The specification of FUNCTN is:

```
double precision FUNCTION FUNCTN(NDIM, X)
  INTEGER                                NDIM
  double precision                      X(NDIM)
```

1: NDIM – INTEGER

Input

On entry: n , the number of dimensions of the integral.

2: X(NDIM) – *double precision* array*Input*

On entry: the co-ordinates of the point at which the integrand f must be evaluated.

FUNCTN must be declared as EXTERNAL in the (sub)program from which D01GBF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

7: EPS – *double precision**Input*

On entry: the relative accuracy required.

Constraint: $\text{EPS} \geq 0.0$.

8: ACC – *double precision**Output*

On exit: the estimated relative accuracy of FINEST.

9: LENWRK – INTEGER

Input

On entry: the dimension of the array WRKSTR as declared in the (sub)program from which D01GBF is called.

For maximum efficiency, LENWRK should be about

$$3 \times \text{NDIM} \times (\text{MAXCLS}/4)^{1/\text{NDIM}} + 7 \times \text{NDIM}.$$

If LENWRK is given the value $10 \times \text{NDIM}$ then the subroutine uses only one iteration of a crude Monte Carlo method with MAXCLS sample points.

Constraint: $\text{LENWRK} \geq 10 \times \text{NDIM}$.

10: WRKSTR(LENWRK) – *double precision* array *Input/Output*

On entry: if MINCLS < 0, WRKSTR must be unchanged from the previous call of D01GBF – except that for a new integrand WRKSTR(LENWRK) must be set to 0.0. See also MINCLS.

On exit: contains information about the current sub-interval structure which could be used in later calls of D01GBF. In particular, WRKSTR(*j*) gives the number of sub-intervals used along the *j*th co-ordinate axis.

11: FINEST – *double precision* *Input/Output*

On entry: must be unchanged from a previous call to D01GBF.

On exit: the best estimate obtained for the integral.

12: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq$ 0 on exit, the recommended value is –1. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or –1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NDIM < 1,
or MINCLS $\geq$ MAXCLS,
or LENWRK < $10 \times \text{NDIM}$,
or MAXCLS < $4 \times (\text{NDIM} + 1)$,
or EPS < 0.0.

IFAIL = 2

MAXCLS was too small for D01GBF to obtain the required relative accuracy EPS. In this case D01GBF returns a value of FINEST with estimated relative error ACC, but ACC will be greater than EPS. This error exit may be taken before MAXCLS nonzero integrand evaluations have actually occurred, if the routine calculates that the current estimates could not be improved before MAXCLS was exceeded.

7 Accuracy

A relative error estimate is output through the parameter ACC. The confidence factor is set so that the actual error should be less than ACC 90% of the time. If you want a higher confidence level then a smaller value of EPS should be used.

8 Further Comments

The running time for D01GBF will usually be dominated by the time used to evaluate FUNCTN, so the maximum time that could be used is approximately proportional to MAXCLS.

For some integrands, particularly those that are poorly behaved in a small part of the integration region, D01GBF may terminate with a value of ACC which is significantly smaller than the actual relative error. This should be suspected if the returned value of MINCLS is small relative to the expected difficulty of the integral. Where this occurs, D01GBF should be called again, but with a higher entry value of MINCLS (e.g., twice the returned value) and the results compared with those from the previous call.

The exact values of FINEST and ACC on return will depend (within statistical limits) on the sequence of random numbers generated within D01GBF by calls to G05SAF. Separate runs will produce identical answers.

9 Example

This example calculates the integral

$$\int_0^1 \int_0^1 \int_0^1 \int_0^1 \frac{4x_1x_3 \exp(2x_1x_3)}{(1+x_2+x_4)^2} dx_1 dx_2 dx_3 dx_4 = 0.575364.$$

9.1 Program Text

```
*      D01GBF Example Program Text
*      Mark 14 Revised. NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NDIM, MAXCLS, LENWRK
      PARAMETER        (NDIM=4,MAXCLS=20000,LENWRK=500)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*      .. Local Scalars ..
      DOUBLE PRECISION ACC, EPS, FINEST
      INTEGER          IFAIL, K, MINCLS
*      .. Local Arrays ..
      DOUBLE PRECISION A(NDIM), B(NDIM), WRKSTR(LENWRK)
*      .. External Functions ..
      DOUBLE PRECISION FUNCTN
      EXTERNAL         FUNCTN
*      .. External Subroutines ..
      EXTERNAL         D01GBF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'D01GBF Example Program Results'
      DO 20 K = 1, NDIM
        A(K) = 0.0D0
        B(K) = 1.0D0
20    CONTINUE
      EPS = 0.01D0
      MINCLS = 1000
      IFAIL = 1
*
      CALL D01GBF(NDIM,A,B,MINCLS,MAXCLS,FUNCTN,EPS,ACC,LENWRK,WRKSTR,
+              FINEST,IFAIL)
*
      WRITE (NOUT,*)
      IF (IFAIL.GT.0) THEN
        WRITE (NOUT,99999) 'D01GBF fails. IFAIL =', IFAIL
        WRITE (NOUT,*)
      END IF
      IF (IFAIL.EQ.0 .OR. IFAIL.EQ.2) THEN
        WRITE (NOUT,99998) 'Requested accuracy      = ', EPS
        WRITE (NOUT,99997) 'Estimated value       = ', FINEST
        WRITE (NOUT,99998) 'Estimated accuracy     = ', ACC
        WRITE (NOUT,99999) 'Number of evaluations = ', MINCLS
      ELSE IF (IFAIL.LT.0) THEN
        WRITE (NOUT,99999) ' ** D01GBF returned with IFAIL = ', IFAIL
      END IF
*
```

```

99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,A,E13.2)
99997 FORMAT (1X,A,F13.5)
      END
*
      DOUBLE PRECISION FUNCTION FUNCTN(NDIM,X)
*      .. Scalar Arguments ..
      INTEGER          NDIM
*      .. Array Arguments ..
      DOUBLE PRECISION X(NDIM)
*      .. Intrinsic Functions ..
      INTRINSIC        EXP
*      .. Executable Statements ..
      FUNCTN = 4.0D0*X(1)*X(3)**2*EXP(2.0D0*X(1)*X(3))/(1.0D0+X(2)+X(4))
+          **2
      RETURN
      END

```

9.2 Program Data

None.

9.3 Program Results

D01GBF Example Program Results

Requested accuracy	=	0.10E-01
Estimated value	=	0.57572
Estimated accuracy	=	0.92E-02
Number of evaluations	=	1728
